
HP-33S Calculator Program Closure 4

Three-Point Horizontal Resection Reduction Program

Programmer: Dr. Bill Hazelton

Date: December, 2005.

Line Instruction Display User Programming Instructions
R0001 LBL R
R0002 SF 10  FLAGS  SF  .0
R0003 ENTER LEFT X  EQN  RCL E  RCL N etc. ENTER to end
R0004 PSE
R0005 INPUT  X
R0006 STO  A
R0007 ENTER LEFT Y  EQN  RCL E  RCL N etc. ENTER to end
R0008 PSE
R0009 INPUT  Y
R0010 STO  B
R0011 ENTER MID X  EQN  RCL E  RCL N etc. ENTER to end
R0012 PSE
R0013 INPUT  X
R0014 STO  C
R0015 ENTER MID Y  EQN  RCL E  RCL N etc. ENTER to end
R0016 PSE
R0017 INPUT  Y
R0018 STO  D
R0019 ENTER RIGHT X  EQN  RCL E  RCL N etc. ENTER to end
R0020 PSE
R0021 INPUT  X
R0022 STO  E
R0023 ENTER RIGHT Y  EQN  RCL E  RCL N etc. ENTER to end
R0024 PSE
R0025 INPUT  Y
R0026 STO  F
R0027 ENTER ALPHA  EQN  RCL E  RCL N etc. ENTER to end
R0028 PSE
R0029 INPUT  X
R0030 →HR
R0031 STO  G
R0032 ENTER BETA  EQN  RCL E  RCL N etc. ENTER to end
R0033 PSE
R0034 INPUT  X
R0035 →HR
R0036 STO  H
R0037 RCL  A
R0038 RCL–  C
R0039 RCL  B

R0040 RCL–  D
R0041 → θ, r  4
R0042 STO  L
R0043 x < > y
R0044 STO  M
R0045 RCL  E
R0046 RCL–  C
R0047 RCL  F
R0048 RCL–  D
R0049 → θ, r  4
R0050 STO  K
R0051 x < > y
R0052 STO  N
R0053 360
R0054 STO  Z
R0055 RCL  M
R0056 RCL–  N
R0057 x < 0 ?
R0058 RCL+  Z
R0059 STO  I
R0060 RCL+  G
R0061 RCL+  H
R0062 RCL  Z
R0063 x < > y
R0064 –
R0065 STO  S
R0066 RCL  L
R0067 RCL  H
R0068 SIN
R0069 ×
R0070 RCL÷  K
R0071 RCL  G
R0072 SIN
R0073 ÷
R0074 RCL  S
R0075 SIN
R0076 ÷
R0077 RCL  S
R0078 TAN
R0079 1/x
R0080 +
R0081 1/x
R0082 ATAN
R0083 STO  X
R0084 RCL  M
R0085 180
R0086 +

R0087 RCL+  G
R0088 RCL+  X
R0089 STO  Y
R0090 RCL  L
R0091 RCL  X
R0092 SIN
R0093 ×
R0094 RCL  G
R0095 SIN
R0096 ÷
R0097 RCL  Y
R0098 x < > y
R0099 → y, x  4
R0100 RCL+  D
R0101 STO  Y
R0102 x < > y
R0103 RCL+  C
R0104 STO  X
R0105 UNKNOWN X =  EQN  RCL U  RCL N etc. ENTER to end
R0106 PSE
R0107 VIEW  X
R0108 UNKNOWN Y =  EQN  RCL U  RCL N etc. ENTER to end
R0109 PSE
R0110 VIEW Y
R0111 RCL  I
R0112 RCL+  G
R0113 RCL+  H
R0114 →HMS
R0115 CHECK VALUE  EQN  RCL C  RCL H etc. ENTER to end
R0116 PSE
R0117 STOP
R0118 CF 10  FLAGS  CF  .0
R0119 RTN

Notes

(1) Horizontal 3-point resection solution, based on measuring two angles at an
unknown point to three known points.

(2) Brief prompts are provided before each requirement for data entry, as well as
before results are displayed. The prompt shows for about 1 second, and is then
replaced by the value or request for input.

(3) Co-ordinates of the unknown point are displayed following brief prompts. They
are also stored in registers for later retrieval.

(4) Angles are entered and displayed in HP notation, i.e., DDD.MMSS. Internal
storage of angles and bearings is in decimal degrees.

Theory

This 2-D resection uses Ormsby’s solution. In the discussion below, A is the left point, B is the
middle point, C is the right point, and P is the unknown point. The left angle is alpha (α) and the
right angle is beta (β). The interior angle at B is gamma (γ). The angle at point A is x, which is
the first objective of the solution.

A

B

C

P

x y

α β

γ

α and β are angles observed from the point P to points A, B and C, whose co-ordinate are known.

BP  =  
AB sin x

sin α     =  
BC sin y

sin β   

and (x + y)  =  (360° – (α + β + γ))  =  s

AB
sin α   sin x  =  

BC
sin β   sin (s – x)  =  

BC
sin β (sin s cos x – cos s sin x)  

AB
sin α   sin x  =  

BC
sin β   sin s cos x – 

BC
sin β   cos s sin x

sin x (
AB

sin α    +  
BC

sin β   cos s)  =  
BC

sin β   sin s cos x

( AB
sin α    +  

BC
sin β   cos s) 

sin β
BC sin s    =  cot x

AB sinβ
BC sin α sin s    +  

BC cos s sin β
BC sin s sin β     =  cot x

AB sin β
BC sin α sin s    +  cot s  =  cot x [this is the equation solved first]

y  =  s  –  x

With x and y determined, the sides AP, BP and CP can be calculated and hence the co-ordinates
of P, as follows:

The azimuth of BP (AzBP) can be determined using   AzBP  =  AzAB  +  α  +  x

The length of BP can be determined using     BP  =  
AB sin x

sin α   

Knowing the co-ordinates of B, AzBP and BP, the co-ordinates of P can be easily computed. As a
check, the equivalent solution can be obtain through the sides AP or CP, or using the angle y.
Note that if P is close the danger circle, a solution will still be obtained, but the sum of α + β + γ
will be close to 180°, probably in the range 175° to 185°. In this case, the solution will be highly
sensitive to changes in α and β. If the solution is close to the danger circle, recomputed with the
angles changed by about their precision and see how much the resulting co-ordinates change. It
can be quite surprising!

Whole circle bearings in HP notation are used. Arbitrary co-ordinates are satisfactory. Plane
surveying assumptions apply. The program uses no error checking on entered data. A check is
made by showing the sum  α + β + γ. If this is close to 180°, the unknown point lies close to the
danger circle and the result is highly suspect.

Sample Computation

Known Points

Point Name X Y

Point A –25.336 778.136

Point B –27.465 1179.927

Point C –30.297 1555.643

Angles Left (α)  =  136° 35' 26"

Right (β)  =  27° 19' 24"

Results Unknown Point (P)  X Co-ordinate  =      26.009

Unknown Point (P)  Y Co-ordinate  =  1101.818

Check Angle  =  344° 02' 32"

Storage Registers Used

A Left known point – X co-ordinate

B Left known point – Y co-ordinate
C Middle known point – X co-ordinate
D Middle known point – Y co-ordinate
E Right known point – X co-ordinate
F Right known point – Y co-ordinate
G Left measured angle — alpha (α)
H Right measured angle — beta (β)
I Interior angle at Middle known point — gamma (γ)
K Distance middle to right point
L Distance middle to left point
M Bearing of middle to left point in decimal degrees
N Bearing of middle to right point in decimal degrees
S s = x + y in decimal degrees
X Initial inputs, then angle x, then X co-ordinate of unknown point
Y Initial inputs, then bearing from middle to unknown point, then Y co-ordinate of

unknown point
Z 360

Labels Used

Label R Length  =  503 Checksum  =  212C

Use the length (LN=) and Checksum (CK=) values to check if program was entered correctly.
Use the sample computation to check proper operation after entry.

Running the Program

Press  XEQ  R

Prompt  ENTER LEFT X briefly, then X?

Enter X Xo-ordinate for left known point.

Press R/S.

Prompt  ENTER LEFT Y briefly, then Y?

Enter Y Xo-ordinate for left known point.

Press R/S.

Prompt  ENTER MID X briefly, then X?

Enter X Xo-ordinate for middle known point.

Press R/S.

Prompt  ENTER MID Y briefly, then Y?

Enter Y Xo-ordinate for middle known point.

Press R/S.

Prompt  ENTER RIGHT X briefly, then X?

Enter X Xo-ordinate for right known point.

Press R/S.

Prompt  ENTER RIGHT Y briefly, then Y?

Enter Y Xo-ordinate for right known point.

Press R/S.

Prompt  ENTER ALPHA briefly, then X?

Enter left angle (α) in HP notation.

Press R/S.

Prompt  ENTER BETA briefly, then X?

Enter right angle (β) in HP notation.

Press R/S.

RUNNING……

Prompt  UNKNOWN X briefly, then X=

X co-ordinate of unknown point (P) is displayed.

Press R/S.

Prompt  UNKNOWN Y briefly, then Y=

Y co-ordinate of unknown point (P) is displayed.

Press R/S.

Prompt  CHECK VALUE briefly.

Sum α + β + γ is displayed in lower line of display in HP notation.

Check that value is not too close to 180°. At least 5 ° away, preferably 15* or more away.

Press R/S to clear flags. Program ends.

Sample Computation 2

Known Points

Point Name X Y

Point A 133.639 1548.712

Point B 158.065 1492.276

Point C 150.267 1353.056

Angles Left (α)  =  5° 01' 48"

Right (β)  =  3° 41' 29"

Results Unknown Point (P)  X Co-ordinate  =   116.784

Unknown Point (P)  Y Co-ordinate  =  1186.818

Check Angle  =  162° 06' 44"

This is not the ideal arrangement for a resection, as the measured angles are quite small. But the
program will still produce an acceptable result.

This example is provided because the other example has negative co-ordinates and this tends to
increase the chances of incorrect data entry. It happened to me, twice!


